

Supplemental Information

Using resonant energy X-ray diffraction to extract order parameter in ternary semiconductors

R. R. Schnepf^{1,2}, B. L. Levy-Wendt^{3,4}, M. B. Tellekamp², B.R. Ortiz², C. L. Melamed^{1,2}, L. T. Schelhas⁴, K. H. Stone⁴, M. F. Toney⁴, E. S. Toberer^{1,2}, and A. C. Tamboli^{1,2}

¹Colorado School of Mines

²National Renewable Energy Laboratory

³Stanford University

⁴SLAC National Accelerator Laboratory

Rietveld refinements were performed on REXD data using the chalcopyrite structure as the base structure for the refinement, see S1 for the reference XRD pattern. A zoomed in view of the peak splitting illustrated in Figure 1 (b) of the main paper is shown in Figure S2. First, the off-resonant full pattern was fit to extract lattice parameters, the phosphorous atomic position, thermal parameter, and texturing. The thermal parameter refined to near zero for both samples; while this is not a physical thermal parameter value, it is likely correcting for another parameter that is changing with Q, such as surface roughness or texturing that was not fully accounted for with the March-Dollase model. In order to minimize the number of parameters we fit, we let the thermal parameter refine to zero rather than introduce another fit parameter. An additional set of lattice parameters also ended up being necessary to achieve correct peak positions for the scans over single peaks as a function of energy; this is probably due to small alignment errors in those scans. The occupancy values were determined by simultaneously refining the full pattern scans and the scans over both the superlattice, (101), and fundamental, (204), peaks with changing energy. Figure S3 shows a zoomed in view of the (101) peaks from the full patterns in Figure 2 of the main paper.




Figure S1. The XRD patterns for the chalcopyrite and zincblende structures for ZnGeP_2 are shown.




Figure S2. A zoomed-in view of the peak splitting that is associated with the chalcopyrite distortion and shown in Fig.1 (b) of the main paper.




Figure S3. Zoomed-in views of the (101) peak from the survey scans (a) for the sample annealed at 650°C (b) for the sample annealed at 600°C.

To ensure that final occupancy values correspond to the absolute lowest R-value, R-values were collected while cycling the initial guesses for the occupancies from fully ordered ($S=1$) to fully disordered ($S=0$). For both samples, the lowest R-values correspond to an order parameter that is near the final solution, corresponding to the lowest R-value. The results of cycling through the initial guess for each sample are shown in figures S4 and S5. The final occupancy values are shown in Table 1.




Figure S4. (a) Overall R_{wp} is plotted against the initial guesses for S ranging from fully disordered to ordered for the sample annealed at 600°C. (b) S is plotted against initial guess and it can be

seen that the lowest R_{wp} values correspond to an S value near 0.32, which is the value that corresponds to the overall minimum in R_{wp} .




Figure S5. (a) The overall R_{wp} is plotted against the initial guesses for S ranging from fully disordered to ordered for the sample annealed at 650°C. The lowest R -value is at an initial guess of $S=0.51$. (b) S is plotted against initial guess and it can be seen that the lowest R_{wp} values correspond to an S value near 0.51.

Sample	c/a	Zn_{Zn} occ.	Ge_{Ge} occ.	S
Annealed at 650°C	1.963(1)	0.76(6)	0.74(5)	0.51(1)
Annealed at 600°C	1.968(3)	0.66(9)	0.65(1)	0.32(0)

Table 1. c/a , the occupancy values, and s determined from each refinement are listed for the corresponding samples.

Topas Academic-64 V6 and jEdit 4.3.1 were used to do the Rietveld refinements. An example input file is shown below:

```

-----
'Input file for Rietveld Refinement of REXD data on ZB052a
'Measured at SSRL 2-1 in March 2019
'Refinement by Rekha Schnepf and Ben Levy-Wendt
-----

r_wp R_wp_overall 7.76660166 r_exp 8.38159847 r_p 4.53594337 r_wp_dash 18.6450434 r_p_dash 14.017523
r_exp_dash 20.1214475 weighted_Durbin_Watson 522.350466 gof 0.926625355

-----
'General information about refinement here
-----

iters 100000
chi2_convergence_criteria 0.001
do_errors
'use the following lines in order to perform 100 cycles
'also output the refined parameters and Rwp values for each cycle
'continue_after_convergence
'prm dummy 0.00000 val_on_continue = If(Cycle == 100, Get(iters) = 0, 0);
'out_prm_vals_on_convergence occ_vals_per_cycle.txt

```

```

'-----
'Define all relevant paramaters here
'-----

prm !Deg2Rad = Pi/180; : 0.01745
prm !w 1.00000
prm !thick .00003
prm !overallb 0.00006_0.06271_LIMIT_MIN_0 min 0 max 10
prm !offset 3.66931_0.19637 min -20 max 20
prm !p_pos 0.25941_0.00343
prm !texture 0.61134_0.00202 'see bottom of input file for texturing macro
prm !cell_a1 5.44651_0.00006 '101 & 204 lattice params
prm !cell_c1 10.69464_0.00014 '101 & 204 lattice params
prm !cell_a2 5.45346_0.00016 'full pattern lattice params
prm !cell_c2 10.71064_0.00040 'full pattern lattice params
prm !strainL1 0.54731_0.00278 'full pattern peak shape
prm !strainG1 0.43936_0.00327 'full pattern peak shape
prm !strainL2 0.38831_0.00120 min .001 max 2 '204 peak shape
prm !strainG2 0.51281_0.00099 min .001 max 2 '204 peak shape
prm !strainL3 0.44946_0.06856 min .001 max 4 '101 gepeak shape
prm !strainG3 1.14300_0.04788 min .001 max 2 '101 ge peak shape
prm !strainL4 0.00498 min .001 max 4 '101 Zn peak shape
prm !strainG4 1.26199 min .001 max 2 '101 Zn peak shape
prm oZn1Ge 0.76626`_0.00170 min 0 max 1 'val_on_continue= -(0.5/100)*Cycle + 1;
prm oGeZn1 0.23374`_0.00157_LIMIT_MAX_0.233742029 max=1-oZn1Ge; 'val_on_continue= (0.5/100)*Cycle
+ 0;
prm oGe2Zn 0.74568`_0.00156 min 0 max 1 'val_on_continue= -(0.5/100)*Cycle + 1;
prm oZnGe2 0.25242`_0.00170_LIMIT_MAX_0.254315849 max=1-oGe2Zn; 'val_on_continue= (0.5/100)*Cycle
+ 0;
'-----

'Input all datasets for corefinemnt here
'Include information that is local to the specific dataset
'-----
'-----

'Zinc edge
'-----

xdd surveyZnEdge_scan1_asymmetry_masked.xy
    r_wp 7.93772125
    gof 0.710150375
    x_calculation_step = Yobs_dx_at(Xo); convolution_step 4
    bkg 65.5494216_0.0796310142 -11.7735614_0.129574792 9.54843119_0.126656586 -
8.91543344_0.107232471 2.909101_0.102845312
    local !hv = 9659+offset;
    local cell_a=cell_a2;
    local cell_c=cell_c2;
    lam ymin_on_ymax 0.0001 la 1.0 lo =12398.42/hv; lh 0.1
    'input info for chaclopyrite structure here
    str
        scale ls_zn_edge 6.00837406e-005_1.365e-007
        local strainL = strainL1;
        local strainG = strainG1;
        local LAC=Phase_LAC_Eqn_1_on_cm; :468.52059
        local mixLAC = Get(mixture_MAC) Get(mixture_density_g_on_cm3); : 468.52059

```

```

PO_asymmetric_reflection(r, =texture; , lomega, 1, 1 1 2)
'-----
'off-res
'-----
xdd ZB053_7_surveyOffRes10500_scan1.xy
    prm !s_off_res 5.70192872e-005_1.165e-007
    r_wp 2.86972277
    gof 0.486411043
    x_calculation_step = Yobs_dx_at(Xo); convolution_step 4
    bkg 263.724279_0.155620558 -16.8209569_0.252105793 16.0375858_0.244217289 -
11.5301356_0.208647337 6.77470536_0.205086659
    local !hv = 10500+offset;
    local cell_a=cell_a2;
local cell_c=cell_c2;
    lam ymin_on_ymax 0.0001 la 1.0 lo =12398.42/hv; lh 0.1
    'finish_X 50
    'input info for chaclopyrite structure here
    str
        scale = s_off_res;
        local strainL = strainL1;
        local strainG = strainG1;
        local LAC=Phase_LAC_Eqn_1_on_cm; :378.14317
    local mixLAC = Get(mixture_MAC) Get(mixture_density_g_on_cm3); : 378.14317
    PO_asymmetric_reflection(r, =texture; , lomega, 1, 1 1 2)
'-----
'Ge edge
'-----
xdd zb053_7_surveyGeedge_scan1.xy
    r_wp 2.07229333
    gof 0.47461545
    x_calculation_step = Yobs_dx_at(Xo); convolution_step 4
    bkg 507.828651_0.215532897 -12.0838955_0.34900238 24.4129654_0.337285535 -
10.0338279_0.288028844 10.6036345_0.285046513
    local !hv = 11103+offset;
    local cell_a=cell_a2;
local cell_c=cell_c2;
    lam ymin_on_ymax 0.0001 la 1.0 lo =12398.42/hv; lh 0.1
    'input info for chaclopyrite structure here
    str
        scale !s_ge_edge 5.83016434e-005_1.443e-007
        local strainL = strainL1;
        local strainG = strainG1;
        local LAC=Phase_LAC_Eqn_1_on_cm; :578.76506
    local mixLAC = Get(mixture_MAC) Get(mixture_density_g_on_cm3); : 578.76506
    PO_asymmetric_reflection(r, =texture; , lomega, 1, 1 1 2)
'-----
'204 peaks
'-----
'Input all 204 datasets example below
xdd "a2scan_GeEdge_204_scan1_11004p12.xye"
    prm !s_11004p12 0.38867_0.00384
    r_wp R_wp_val_204_11004p12 7.6606513
    gof 0.63891132

```

```

x_calculation_step = Yobs_dx_at(Xo); convolution_step 4
bkg 10.4923038_0.322167934 0.126295342_0.371053921 -2.07561946_0.559271543
local cell_a=cell_a1;
local cell_c=cell_c1;
local !hv = 11004.12+offset;
lam ymin_on_ymax 0.0001 la 1.0 lo =12398.42/hv; lh 0.1
str
    scale= s_11004p12 s_off_res;
    local strainL = strainL2;
    local strainG = strainG2;
    local LAC=Phase_LAC_Eqn_1_on_cm; :334.41436
    local mixLAC = Get(mixture_MAC) Get(mixture_density_g_on_cm3); : 334.41436
    PO_asymmetric_reflection(r, =texture; , !omega, 1, 1 1 2)
xdd_out ##INP_File##_204_11004p12.txt load out_record out_fmt out_eqn {
    "%11.5f " = X;
    "%11.5f " = Yobs;
    "%11.5f " = Ycalc;
    "%11.5f" = Yobs-Ycalc;
    "%11.5f\n" = R_wp_val_204_11004p12; }
.
.
.
xdd "a2scan_ZnEdge_204_scan1_9759p9397.xye"
    prm ls_9759p9397 0.46970_0.00457
    r_wp R_wp_val_204_9759p9397 11.7911993
    gof 0.907772227
    x_calculation_step = Yobs_dx_at(Xo); convolution_step 4
    bkg 25.7191972_0.268511626 1.01314243_0.388899619 0.4582185_0.455743784
    local cell_a=cell_a1;
    local cell_c=cell_c1;
    local !hv = 9759.9397+offset;
    lam ymin_on_ymax 0.0001 la 1.0 lo =12398.42/hv; lh 0.1
    str
        scale= s_9759p9397 s_off_res;
        local strainL = strainL2;
        local strainG = strainG2;
        local LAC=Phase_LAC_Eqn_1_on_cm; :458.57049
        local mixLAC = Get(mixture_MAC) Get(mixture_density_g_on_cm3); : 458.57049
        PO_asymmetric_reflection(r, =texture; , !omega, 1, 1 1 2)
    xdd_out ##INP_File##_204_9759p9397.txt load out_record out_fmt out_eqn {
        "%11.5f " = X;
        "%11.5f " = Yobs;
        "%11.5f " = Ycalc;
        "%11.5f" = Yobs-Ycalc;
        "%11.5f\n" = R_wp_val_204_9759p9397; }
'-----
'101 peaks
'-----
'Input all 101 peaks
xdd "a2scan_GeEdge_101_2_scan1_11004p12.xye"
    weighting = 10/SigmaYobs^2;
    r_wp R_wp_val_101_11004p12 11.0600217
    gof 1.45523115

```

```

start_X 13.025064498938985
finish_X 13.800950293641872
x_calculation_step = Yobs_dx_at(Xo); convolution_step 4
bkg 16.0162439_0.132276119 -2.51494231_0.220775847
local cell_a=cell_a1;
local cell_c=cell_c1;
local !hv = 11004.12+offset;
lam ymin_on_ymax 0.0001 la 1.0 lo =12398.42/hv; lh 0.1
str
    scale= s_11004p12 s_off_res;
    local strainL = strainL3;
    local strainG = strainG3;
    local LAC=Phase_LAC_Eqn_1_on_cm; :334.41436
    local mixLAC = Get(mixture_MAC) Get(mixture_density_g_on_cm3); : 334.41436
    PO_asymmetric_reflection(r, =texture; , !omega, 1, 1 1 2)
xdd_out ##INP_File##_101_11004p12.txt load out_record out_fmt out_eqn {
    " %11.5f " = X;
    " %11.5f " = Yobs;
    " %11.5f " = Ycalc;
    " %11.5f" = Yobs-Ycalc;
    " %11.5f\n" = R_wp_val_101_11004p12;}
.
.
.
xdd "a2scan_ZnEdge_101_scan1_9759p9397.xye"
    weighting = 10/SigmaYobs^2;
    r_wp R_wp_val_101_9759p9397 8.31711783
    gof 1.49667935
    start_X 14.69411171222775
    finish_X 15.5705514323667
    x_calculation_step = Yobs_dx_at(Xo); convolution_step 4
    bkg 29.6091272_0.172047322 -0.0810908_0.286365243
    local cell_a=cell_a1;
    local cell_c=cell_c1;
    local !hv = 9759.9397+offset;
    lam ymin_on_ymax 0.0001 la 1.0 lo =12398.42/hv; lh 0.1
    str
        scale= s_9759p9397 s_off_res;
        local strainL = strainL3;
        local strainG = strainG3;
        local LAC=Phase_LAC_Eqn_1_on_cm; :458.57049
        local mixLAC = Get(mixture_MAC) Get(mixture_density_g_on_cm3); : 458.57049
        PO_asymmetric_reflection(r, =texture; , !omega, 1, 1 1 2)
    xdd_out ##INP_File##_101_9759p9397.txt load out_record out_fmt out_eqn {
        " %11.5f " = X;
        " %11.5f " = Yobs;
        " %11.5f " = Ycalc;
        " %11.5f" = Yobs-Ycalc;
        " %11.5f\n" = R_wp_val_101_9759p9397;}

```

```

'-----
'Information on datasets that is general to all
'Type in phase/space group/cell etc
'-----

```

```

for xdds 1 to 395 {
  LP_Factor( 90)
  Zero_Error(!zero, 0.01992_0.00043)
  for strs 1 to 1 {
    phase_name Chalc
    Tetragonal( =cell_a; , =cell_c;)
    volume 317.251
    space_group "I-42d"
    'site Zn1
    site Zn1Ge x =0; y =0; z =0; occ Zn =oZn1Ge; min 0
max 1
    beq = overallb;
    site GeZn1 x =0; y =0; z =0; occ Ge
=oGeZn1; beq = overallb;
    'site Ge1
    site Ge2Zn x =0; y =0; z =1/2; occ Ge =oGe2Zn;
min 0 max 1
    beq = overallb;
    site ZnGe2 x =0; y =0; z =1/2; occ Zn
=oZnGe2; beq = overallb;
    site P1 x= p_pos; y =1/4; z =1/8;
occ P 1
    beq = overallb;
    r_bragg 5.59638721
    Phase_Density_g_on_cm3( 4.18361)
    Strain_L(, strainL)
    Strain_G(, strainG)
    'penalty functions for constraining with composition from XRF
    penalties_weighting_K1 4
    penalty = (.5(oZn1Ge+oZnGe2)-.51)^2; : 4.38357874e-007`
    penalty = (.5(oGe2Zn+oGeZn1)-.49)^2; : 8.23173065e-008`
  }
}
'-----
'define the march-dollase macro to account for texturing
'-----
macro PO_asymmetric_reflection(r_c, r_v, omega_c, omega_v, hkl) {
  #m_argu omega_c
  If_Prm_Eqn_Rpt(omega_c, omega_v, min 0.0001 max 90) 'incident angle in degrees'
  prm #m_unique Delta = Abs(Th - ((CeV(omega_c, omega_v))*Deg)) ;
  generalised_PO_eqn(r_c, r_v, , hkl, , Delta, 16)
}
macro & cosrho(& alpha, & Delta, & phi) {
  Cos(alpha)*Cos(Delta) - Sin(alpha)*Sin(Delta)*Sin(phi)
}
macro & PO_P(& r, & alpha, & Delta, & phi) {
  (r^2 * cosrho(alpha, Delta, phi)^2 + (1-cosrho(alpha, Delta, phi)^2)/r)^(-3/2)
}
macro & PO_f(& r, & alpha, & Delta, & N) {
  'hard coded this for N==16 as the Sum formalism is not working '
  '(1/N)*Sum( PO_P(r, alpha, Delta, (j+(1/2))*Pi/N) , j=0, j<=N-1, j=j+1)'
  (1/16)*
  (PO_P(r, alpha, Delta, ( 0+(1/2))*Pi/16) +
  PO_P(r, alpha, Delta, ( 1+(1/2))*Pi/16) +
  PO_P(r, alpha, Delta, ( 2+(1/2))*Pi/16) +
  PO_P(r, alpha, Delta, ( 3+(1/2))*Pi/16) +

```

```

        PO_P(r,alpha,Delta,( 4+(1/2))*Pi/16) +
        PO_P(r,alpha,Delta,( 5+(1/2))*Pi/16) +
        PO_P(r,alpha,Delta,( 6+(1/2))*Pi/16) +
        PO_P(r,alpha,Delta,( 7+(1/2))*Pi/16) +
        PO_P(r,alpha,Delta,( 8+(1/2))*Pi/16) +
        PO_P(r,alpha,Delta,( 9+(1/2))*Pi/16) +
        PO_P(r,alpha,Delta,(10+(1/2))*Pi/16) +
        PO_P(r,alpha,Delta,(11+(1/2))*Pi/16) +
        PO_P(r,alpha,Delta,(12+(1/2))*Pi/16) +
        PO_P(r,alpha,Delta,(13+(1/2))*Pi/16) +
        PO_P(r,alpha,Delta,(14+(1/2))*Pi/16) +
        PO_P(r,alpha,Delta,(15+(1/2))*Pi/16)
    }
'angles in radians'
macro generalised_PO_eqn(r_c, r_v, alpha, hkl, Delta_c, Delta_v, N) {
    #m_argu r_c
    #m_argu Delta_c
    If_Prm_Eqn_Rpt(r_c, r_v, min 0.0001 max = 2 Val + .5;) 'preferred orientation parameter'
    If_Prm_Eqn_Rpt(Delta_c, Delta_v, min 0.0001 max =Pi;) 'angle in radians: ==0 symmetric reflection, ==Pi/2
    capillary transmission, ==Abs(Th-Omega) asymmetric reflection'
    'The following code modified from S11.5.4 in the TechRef & PO macro in topas.inc'
    #m_ifarg alpha ""
    #m_unique_not_refine alpha
    #m_endif
    str_hkl_angle alpha hkl 'angle in radians between PO direction and current diffraction vector'
    scale_pks = Multiplicities_Sum( PO_f(CeV(r_c,r_v), alpha, CeV(Delta_c,Delta_v), N) );
}

'-----
'Create a text file that outputs all of the relevant paramaters
'-----

out paramater_outputs.txt append
'first add the headers
Out_String ("R_wp_overall" )
Out_String ("    overallb overallb_error")
Out_String ("    offset    offset_error ")
Out_String ("    cell_a1    cell_a1_error ")
Out_String ("    cell_c1    cell_c1_error ")
Out_String ("    cell_a2    cell_a2_error ")
Out_String ("    cell_c2    cell_c2_error ")
Out_String ("    strainL1 strainL1_error")
Out_String ("    strainG1 strainG1_error")
Out_String ("    strainL2 strainL2_error")
Out_String ("    strainG2 strainG2_error")
Out_String ("    p_pos    p_pos_error ")
Out_String ("    texture texture_error ")
Out_String ("    oZn1Ge oZn1Ge_error ")
Out_String ("    oGeZn1 oGeZn1_error ")
Out_String ("    oGe2Zn oGe2Zn_error ")
Out_String ("    oZnGe2 oZnGe2_error\n ")
'then add the actual values
Out (R_wp_overall, " %11.5f")
Out (overallb, " %11.5f", " %11.5f")

```

```
Out (offset , "%11.5f", "%11.5f")
Out (cell_a1 , "%11.5f", "%11.5f")
Out (cell_c1 , "%11.5f", "%11.5f")
Out (cell_a2 , "%11.5f", "%11.5f")
Out (cell_c2 , "%11.5f", "%11.5f")
Out (strainL1, "%11.5f", "%11.5f")
Out (strainG1, "%11.5f", "%11.5f")
Out (strainL2, "%11.5f", "%11.5f")
Out (strainG2, "%11.5f", "%11.5f")
Out (p_pos , "%11.5f", "%11.5f")
Out (texture , "%11.5f", "%11.5f")
Out (oZn1Ge , "%11.5f", "%11.5f")
Out (oGeZn1 , "%11.5f", "%11.5f")
Out (oGe2Zn , "%11.5f", "%11.5f")
Out (oZnGe2 , "%11.5f", "%11.5f\n")
```